

## 3.4 Bewegungen auf einer Raumkurve

Geradlinige Bewegungen sind sicherlich nicht die häufigsten Bewegungsabläufe. Praktisch jede Bewegung geschieht entlang einer Kurve im Raum. Schon im letzten Kapitel haben wir spezielle Fälle diskutiert und Ausdrücke wie Krümmung und Krümmungsradius eingeführt. Es handelt sich dabei um Begriffe, die in der Differentialgeometrie entwickelt werden und mit denen wir uns zunächst etwas eingehender befassen müssen.

### 3.4.1 Grundbegriffe

Die Länge des Geschwindigkeitsvektors ist  $v = ds/dt$ , der Vektor  $\vec{v}$  zeigt in Richtung der Bewegung. Es gilt  $v = (\vec{v} \cdot \vec{v})^{1/2}$ , worin  $\vec{v} \cdot \vec{v}$  das Skalarprodukt der Vektoren  $\vec{v}$  und  $\vec{v}$  ist. Der Vektor

$$\vec{t} = \frac{\vec{v}}{v} = \frac{\vec{v}}{\dot{s}} = \frac{\vec{v}}{\frac{ds}{dt}} \quad (3.4-1)$$

ist ein tangentialer Einheitsvektor, er zeigt ebenfalls in Richtung der Bewegung. In der x-y-Ebene kann man schreiben  $\vec{t} = \cos(\phi)\vec{i} + \sin(\phi)\vec{j}$ , worin  $\phi$  der Winkel ist, den  $\vec{v}$  mit der positiven x-Achse einschließt. Man benutzt häufig die Länge des Bogens  $s = s(t)$  als Parameter anstelle von  $t$  (Zeit). In diesem Fall ist  $\vec{r}(s)$  der Positionsvektor, und seine Ableitung bezüglich  $s$  bezeichnet man mit  $d\vec{r}(s)/ds := \vec{r}'(s)$ . Da  $\vec{v}(t) = d\vec{r}(t)/dt = d\vec{r}(s)/ds \cdot d(s)/dt = d\vec{r}(s)/ds \cdot \dot{s} = \vec{r}'(s) v$ , können wir auch schreiben

$$\vec{r}'(s) = \vec{v}(t)/v = \vec{t}(s) \quad (3.4-2)$$

Die Ableitung von  $\vec{t}$  nach  $s$  wird *Krümmung*  $\kappa$  (Kappa) genannt:

$$\kappa = 1/\rho = |d\vec{t}/ds| = |\vec{t}'(s)| = |\vec{r}''(s)| \quad (3.4-3)$$

Der Normalen-Einheitsvektor  $\vec{n}$  ist gegeben durch

$$\vec{n} = \rho \vec{r}''(s) \quad (3.4-4)$$

Ausgehend von (3.4-1) erhalten wir

$$\vec{r}'' = d\vec{r}' = d(\vec{v}(t)/v)/ds = d\vec{t}(t)/ds = d\vec{t}/dt \cdot 1/(ds/dt) = d\vec{t}/dt \cdot 1/(\vec{v} \cdot \vec{v})^{1/2},$$

oder

$$\vec{r}''(s) = d\vec{t}(t)/dt \cdot 1/(\vec{v} \cdot \vec{v})^{1/2} \quad (3.4-5)$$

Hier ist  $(\vec{v} \cdot \vec{v})^{1/2}$  der Betrag  $|\vec{v}| = v$  des Vektors  $\vec{v}$ .

Wir können Kappa auch mit Hilfe des Vektorproduktes berechnen, denn es gilt

$$\kappa = \frac{|\vec{v} \times \vec{a}|}{\left(\sqrt{\vec{v} \cdot \vec{v}}\right)^3} \quad (3.4-6)$$

(In MuPAD gibt es die Prozedur `linalg::crossProduct`).

Für den Krümmungsvektor können wir die folgende Gleichung einsetzen

$$\vec{\kappa} = (\vec{a} - \vec{v} \cdot dv/dt \cdot 1/v)/v^2 \quad (3.4-7)$$

Schließlich haben wir noch eine oft nützliche Gleichung für den Krümmungsradius einer Kurve

$$\rho = (x'^2 + y'^2)^{3/2}/(x'y'' - y'x'') \quad (3.4-8)$$

Im Fall einer Ellipse ( $x = a \cos(t), y = b \sin(t)$ ) benutzen wir

$$\rho = (a^2 \sin^2(t) + b^2 \cos^2(t))^{3/2}/(ab) \quad (3.4-9)$$

(Weiter unten im Abschnitt 3.4.3 werden wir genauer auf die vorhin eingeführten Begriffe eingehen.)

Mit Hilfe der jetzt folgenden MuPAD-Programme werden wir die Anwendung der Formeln demonstrieren.

### 3.4.2 Programme

Im ersten Programm berechnen wir für  $t = \pi/7$  die *Krümmung* und den *Krümmungsradius* einer Ellipse mit Hilfe der Formeln (3.4-5), (3.4-6), (3.4-7) und (3.4-9).

#### Programm 1

```

reset();//Kruemmung mit versch. Formeln
t1:=PI/7:
a:=9:b:=5:c:=0:// Ellipse
x:=t->a*cos(t):
y:=t->b*sin(t):
z:=t->c*t:// eigtl.Darstellung einer Spirale, hier aber c=0
pos:=matrix([[x(t),y(t),z(t)]])://Ortsvektor
v:=matrix([[x'(t),y'(t),z'(t)]])://Geschw.-Vektor
ac:=matrix([[x''(t),y''(t),z''(t)]])://Beschl.-Vektor
absolv:=sqrt(v[1,1]^2+v[1,2]^2+v[1,3]^2)://v-absolut
derv:=diff(absolv,t)://d|v|/dt//Ableitung
vt:=v/absolv://Tangenteneinheitsvektor
vc:=diff(vt,t)/absolv://Kruemmungsvektor (3.4-5)
cross:=linalg::crossProduct(v,ac):
abscross:=sqrt(linalg::scalarProduct(cross,cross)):

```

```

t:=t1:
absolv:=sqrt(vc[1,1]^2+vc[1,2]^2+vc[1,3]^2):
kk:=ac/absolv^2-derv*v/absolv^3://Kruemmungsvektor (3.4-7)
absolk:=float(abcross/absolv^3):(3.4-6)
//Kruemmungsradien
ro:=1/absolk;
ro1:=1/float(absolv);//Kruemmungsradius
ro2:=1/float(sqrt(kk[1,1]^2+kk[1,2]^2+kk[1,3]^2));
ro2d:=float((a^2*sin(t)^2+b^2*cos(t)^2)^1.5/(a*b))

```

Wir erhalten viermal das Ergebnis: 4.708748705

## Programm 2

Das zweite Programm erzeugt eine graphische Darstellung des Krümmungsvektors im Fall einer Ellipse. Beachte die Länge des Vektors  $\vec{r}'$  an verschiedenen Stellen der Kurve.

```

reset();//Kruemmungsvektor bei einer Ellipse
t1:=PI/5:a:=9: b:=5:
x:=t->a*cos(t):
y:=t->b*sin(t):
scale:=20://Ellipse
curve:=plot::Curve2d([x(t),y(t)],t=0..PI):
pos:=matrix([[x(t),y(t)]])://Ortsvektor
v:=matrix([[x'(t),y'(t)]])://Geschw.-Vektor
absolv:=sqrt(v[1,1]^2+v[1,2]^2):
vt:=v/absolv://Tangenteneinheitsvektor
k:=diff(vt,t)/absolv://Kruemmungsvektor
t:=t1:
x1:=pos[1,1]:y1:=pos[1,2]:
x2:=x1+ scale*k[1,1]:
y2:=y1+ scale*k[1,2]:
p:=plot::Point2d(x1,y1,Color=RGB::Green,PointSize=2*unit::mm):
kc:=plot::Arrow2d([x1,y1],[x2,y2],Color=RGB::Red):
plot(curve,p,kc,Scaling=Constrained)

```




Fig.3.4-1

### Programm 3

In diesem Programm erstellen wir eine Animation, die die beiden Vektoren  $\vec{v}$  und  $\vec{r}''$  entlang einer Ellipse laufend zeigt. Beachte, dass  $\vec{v}$  an den Stellen großer Krümmung klein ist (und umgekehrt).

```

reset();//Kruemmung (rot) und Geschwindigkeit (schwarz)
//t1:=PI/2:
a:=9: b:=5:
x:=t->a*cos(t):
y:=t->b*sin(t):
scale:=20://ellipse
curve:=plot::Curve2d([x(t),y(t)],t=0..2*PI):
pos:=matrix([[x(t),y(t)]])://Position
v:=matrix([[x'(t),y'(t)]])://Geschwindigkeit
absolv:=sqrt(v[1,1]^2+v[1,2]^2)//Betrag von v
vt:=v/absolv://Tang.einh.vektor
k:=diff(vt,t)/absolv://Kruemmungsvektor
x1:=pos[1,1]:
y1:=pos[1,2]:
v1:=v[1,1]:
v2:=v[1,2]:
x2:=x1+0.5*v1://Spitze von v mit Massfaktor
y2:=y1+0.5*v2:
x3:=x1+scale*k[1,1]://Spitze von k
y3:=y1+ scale*k[1,2]:
p:=plot::Point2d(x1,y1,t=0..2*PI,Color=RGB::Green,
PointSize=2*unit::mm):
vc:=plot::Arrow2d([x1,y1],[x2,y2],t=0..2*PI,Color=RGB::Black):
kc:=plot::Arrow2d([x1,y1],[x3,y3],t=0..2*PI,Color=RGB::Red):
plot(curve,p,vc,kc,Scaling=Constrained)

```




Fig.3.4-2

#### Programm 4

Das vierte Programm zeigt den *Berührungskreis* in dem Punkt  $5\pi/6$  einer Ellipse.

```

reset();//Kruemmung und Kruemmungskreis
t1:=5*PI/6: a:=3: b:=2:
x:=t->a*cos(t):
y:=t->b*sin(t):
scale:=4;//Ellipse
curve:=plot::Curve2d([x(t),y(t)],t=0..PI):
  pos:=matrix([[x(t),y(t)]])://Position
  v:=matrix([[x'(t),y'(t)]])://Geschw.
  absolv:=sqrt(v[1,1]^2+v[1,2]^2):
  et:=v/absolv://Tang.einh.vektor
  k:=diff(et,t)/absolv://Kruemmungsvektor
  radius:=1/sqrt(k[1,1]^2+k[1,2]^2):
  centrum:=pos+k*radius^2:
t:=t1:
x1:=pos[1,1]:
y1:=pos[1,2]:
x2:=x1+scale*k[1,1]:
y2:=y1+scale*k[1,2]:
x3:=centrum[1,1]:
y3:=centrum[1,2]:
kreis:=plot::Circle2d(radius,[x3,y3],Color=RGB::Black):
pc:=plot::Point2d(x3,y3,Color=RGB::Black,PointSize=2*unit::mm):
p:=plot::Point2d(x1,y1,Color=RGB::Green,PointSize=2*unit::mm):
kc:=plot::Arrow2d([x1,y1],[x2,y2],Color=RGB::Red):
plot(curve,pc,p,kc,kreis,Scaling=Constrained)

```




Fig.3.4-3

### 3.4.3 Mit Bleistift und Papier

#### Geschwindigkeit

Wir schauen uns jetzt ein Objekt an, das die räumliche Kurve  $C$  beschreibt. Im Augenblick  $t$  ist das Objekt im Punkt  $A$  mit dem Ortsvektor  $\vec{r}(t) = x(t)\vec{i} + y(t)\vec{j} + z(t)\vec{k}$ , vgl. Fig.3.4-4, S. 9). Wenn  $t$  sich ändert, beschreibt der Vektor  $\vec{r}$  eine Kurve im Raum, die die Darstellung  $\{x = x(t), y = y(t), z = z(t)\}$  hat. Obgleich die Bewegung entlang des Bogens  $AB = \Delta s$  geschieht, beschreiben wir die Verschiebung mit Hilfe des Vektors  $\vec{AB} = \Delta\vec{r} = \vec{r}(t + \Delta t) - \vec{r}(t)$ . Die mittlere Geschwindigkeit  $\Delta\vec{r}/\Delta t = (\vec{r}(t + \Delta t) - \vec{r}(t))/\Delta t$  ist ein Vektor in Richtung  $\Delta\vec{r}$ . Die momentane Geschwindigkeit ist der Grenzwert

$$\lim_{\Delta t \rightarrow 0} \frac{\Delta\vec{r}}{\Delta t} = \frac{d\vec{r}}{dt} := \vec{v} \quad (1)$$

In der Grenze, wenn  $B$  sehr nahe bei  $A$  ist, fällt  $\Delta\vec{r}$  mit der Tangente in  $A$  zusammen, vgl. Fig. 2.2-6. Wir können Gl.(1) leicht in eine Schreibweise mit  $\Delta s$  umformen:

$$\vec{v} = \lim_{\Delta t \rightarrow 0} \left( \frac{\Delta\vec{r}}{\Delta s} \frac{\Delta s}{\Delta t} \right) = \lim_{\Delta s \rightarrow 0} \frac{\Delta\vec{r}}{\Delta s} \cdot \lim_{\Delta t \rightarrow 0} \frac{\Delta s}{\Delta t} \quad (2)$$

In der Grenze ist der Betrag von  $\Delta\vec{r}$  gleich  $\Delta s$ , d.h. der Faktor

$$\lim_{\Delta s \rightarrow 0} \frac{\Delta\vec{r}}{\Delta s} = \frac{d\vec{r}}{ds} = \vec{r}' = \vec{t}(s) \quad (3)$$

ist ein Einheitsvektor in Richtung der Tangente. Somit dürfen wir Gl. (2) schreiben als  $\vec{v} = \frac{ds}{dt} \vec{t} = v\vec{t}$ . In (3.4-2) hatten wir den Tangenteneinheitsvektor schon als  $\vec{t} = \vec{v}(t)/v(t)$  kennen gelernt.

Die Änderungsgeschwindigkeit von  $\vec{t}$  bezüglich  $s$  ist ein Maß für die Krümmung der Kurve im Punkt  $\vec{r}$ . Für den Vektor der Krümmung, vgl. (3.4-5), erhalten wir dann

$$\vec{\kappa} := \frac{d\vec{t}}{ds} = \vec{r}'' \quad (4)$$

$\vec{\kappa}$  hat in jedem Bahnpunkt die Richtung der Senkrechten (Normalen) an die Bahn. Wenn wir mit  $\vec{n}$  den senkrechten Einheitsvektor bezeichnen (*Hauptnormalenvektor*), haben wir

$$\frac{d\vec{t}}{ds} = \kappa \vec{n} \quad (5)$$

Der Einheitsvektor  $\vec{b} := \vec{t} \times \vec{n}$  ist der *Binormalenvektor* der Kurve.  $\vec{b}$  steht senkrecht auf der von  $\vec{t}$  und  $\vec{n}$  aufgespannten Ebene. Das Tripel  $(\vec{t}, \vec{n}, \vec{b})$  bildet ein orthogonales Koordinatensystem in jedem Punkt der Bahnkurve. Die Ableitungen dieser Vektoren ("begleitendes Dreibein") nach  $s$  sehen folgendermaßen aus

$$\begin{aligned} \frac{d\vec{t}}{ds} &= \kappa \vec{n} \\ \frac{d\vec{n}}{ds} &= \tau \vec{b} - \kappa \vec{t} \\ \frac{d\vec{b}}{ds} &= -\tau \vec{n} \end{aligned} \quad (6-8)$$

$\tau$  (Tau) wird *Windung* oder *Torsion* genannt. Die von  $\vec{t}$  und  $\vec{n}$  aufgespannte Ebene heißt *Schmiegebene*,  $\vec{n}$  und  $\vec{b}$  spannen die sogenannte *Normalebene* auf. Zu Ehren des französischen Mathematikers F. Frenet (1816-1868) heißen die Formeln 6 bis 8 *Frenetsche Formeln*. (Die Methode des begleitenden Dreibeins wurde besonders von französischen Geometern ausgearbeitet.)

### Beschleunigung

Wenn sich die Geschwindigkeit eines Objektes in  $\Delta t$ -Sekunden von  $\vec{v}_1$  auf  $\vec{v}_2$  ändert, so ist seine mittlere Beschleunigung  $\vec{a}_m = (\vec{v}_2 - \vec{v}_1)/\Delta t = \Delta \vec{v}/\Delta t$ . Die momentane Beschleunigung ist

$$\vec{a} := \lim_{\Delta t \rightarrow 0} \frac{\Delta \vec{v}}{\Delta t} = \frac{d^2 \vec{r}}{dt^2} = \ddot{\vec{r}} \quad (9)$$

Der Vektor  $\vec{a}$  zeigt immer in Richtung der konkaven Seite der Kurve und kann in eine tangentielle,  $\vec{a}_t$ , und eine normale Komponente,  $\vec{a}_n$ , zerlegt werden.  $\vec{a}_t$  ist die Ursache für die Änderung des Betrages der Geschwindigkeit,  $\vec{a}_n$  erzeugt die Richtungsänderung.

Wir haben  $\vec{a} = d\vec{v}/dt = d(v\vec{t})/dt = dv/dt \cdot \vec{t} + v \cdot d\vec{t}/dt$ . Wenn die Bahnkurve eine Gerade ist, muss  $d\vec{t}/dt$  null sein, denn  $\vec{t}$  ist in diesem Fall konstant. Wenn die Bahn gekrümmt ist, ändert sich die Richtung von  $\vec{t}$  entlang der Bahn, und  $d\vec{t}/dt$  ist verschieden von null. Wenn  $\phi$  der Winkel zwischen  $\vec{t}$  und der x-Achse ist, dann schließt  $\vec{n}$  mit dieser Achse den Winkel  $\phi + \pi/2$  ein und wir können schreiben

$$\begin{aligned}\vec{t} &= \cos(\phi)\vec{i} + \sin(\phi)\vec{j} \\ \vec{n} &= \cos(\phi + \pi/2)\vec{i} + \sin(\phi + \pi/2)\vec{j} \\ &= -\sin(\phi)\vec{i} + \cos(\phi)\vec{j}\end{aligned}$$

Wir können den Zusammenhang zwischen  $\vec{n}$  und  $\vec{t}$  auch als  $\vec{n} = D \cdot \vec{t}$  darstellen, worin D eine Drehmatrix ist

$$D = \begin{pmatrix} \cos(\phi) & \sin(\phi) \\ -\sin(\phi) & \cos(\phi) \end{pmatrix}$$

(Die Komponente eines Vektors in eine gegebene Richtung ist gleich seiner Projektion auf dieselbe.) Für die Ableitung ergibt sich jetzt  $d\vec{t}/dt = -\sin(\phi)d\phi/dt \cdot \vec{i} + \cos(\phi)d\phi/dt \cdot \vec{j}$ , oder

$$\frac{d\vec{t}}{dt} = \frac{d\phi}{dt} \vec{n} \quad (10)$$

d.h. die Ableitung des Tangenteneinheitsvektors steht senkrecht auf der Kurve. Mit  $d\phi/dt = v/\rho$  erhalten wir

$$\frac{d\vec{t}}{dt} = \frac{v}{\rho} \vec{n} \quad (11)$$

und

$$\vec{a} = \frac{dv}{dt} \vec{t} + \frac{v^2}{\rho} \vec{n} \quad (12)$$

Der erste Term ist die Tangentialbeschleunigung, der zweite ist die Normalbeschleunigung. Im Fall einer gleichförmigen **Kreisbewegung** ist  $s = r\phi$  und  $\omega = d\phi/dt$ . Für die Beschleunigung ergibt das

$$\vec{a}(t) = \omega v \vec{n}(t) = v^2/r \vec{n} = \omega^2 r \vec{n} = -\omega^2 \vec{r}(t)$$

### Krümmung einer Raumkurve

Wir müssen uns nochmals die Gleichung (5) ansehen:  $d\vec{t}/ds = \vec{r}'' = \kappa \vec{n}$  mit  $\vec{t} = d\vec{r}/ds := \vec{r}'$ . Wenn wir dies nach der Zeit ableiten, ergibt sich  $d\vec{t}/dt = d\vec{r}'/dt = d\vec{r}'/ds \cdot ds/dt = \vec{r}'' v$ , d.h.

$$\frac{d\vec{t}}{dt} = \vec{r}'' v \quad (13)$$

woraus sich dann der Krümmungsvektor ergibt:

$$\vec{r}'' = \frac{1}{v} \frac{d\vec{t}}{dt} \quad (14)$$

Im ersten und zweiten Programm haben wir mit dieser Formel den Krümmungsvektor berechnet. Um  $\vec{r}''$  zu berechnen, brauchen wir dann  $v = (v_x^2 + v_y^2 + v_z^2)^{1/2}$  und die Ableitung des Vektors  $\vec{t} = \vec{v}(t)/v(t)$ . In beiden Programmen haben wir dies mit den folgenden Anweisungen erreicht:

```

absolv:=sqrt(v[1,1]^2+v[1,2]^2+v[1,3]^2)//v-absolut
derv:=diff(absolv,t)//d|v|/dt = Ableitung
vt:=v/absolv//Tangenteneinheitsvektor
vc:=diff(vt,t)/absolv//Kruemmungsvektor (3.4-5)

```

Den Vektor  $\vec{r}''$  können wir auch direkt berechnen:

$$\vec{r}'' = \frac{v\dot{\vec{v}} - \dot{v}\vec{v}}{v^2} = \frac{1}{v^2}\vec{a} - \frac{\dot{v}\vec{v}}{v^3} \quad (15)$$

Diese Formel haben wir im ersten Programm benutzt

```

kk:=ac/absolv^2-derv*v/absolv^3//Kruemmungsvektor (3.4-7)

```

Formel (14) ist nicht immer die bequemste zur Berechnung der Krümmung. In der Praxis benutzt man meist den Ausdruck

$$\kappa(t) = \frac{|\vec{a}(t) \times \vec{v}(t)|}{v^3(t)} \quad (16)$$

Wir erhalten diese Gleichung, indem wir (12) mit  $\vec{t} = \vec{v}(t)/v(t)$  multiplizieren:

$$\begin{aligned} \vec{a} \times \vec{v} &= (v'(t)\vec{t}(t) + \kappa(t)v^2(t)\vec{n}(t)) \times v(t)\vec{t}(t) \\ &= v'(t)v\vec{t} \times \vec{t} + \kappa v^3\vec{n} \times \vec{t} \\ &= \kappa v^3\vec{n} \times \vec{t} \end{aligned}$$

Der Betrag ist  $|\vec{a} \times \vec{v}| = \kappa v^3 |\vec{n}| |\vec{t}| \sin(\pi/2) = \kappa v^3$

Schließlich bestimmen wir den Mittelpunkt  $\vec{m}$  des Schmiegekrees für den Punkt A auf der Bahn C.

Wir erkennen aus der Figur, dass  $\vec{m} = \vec{r} + \rho\vec{n}$ .




Fig.3.4-4

